



3 4456 0054036 9

ORNL/TM-9693

ornl**OAK RIDGE
NATIONAL
LABORATORY****MARTIN MARIETTA****ALKACYCL,
A BASIC COMPUTER PROGRAM FOR
THE ANALYSIS OF ALKALI METAL
RANKINE POWER CYCLES****J. C. Moyers**

OAK RIDGE NATIONAL LABORATORY

CENTRAL RESEARCH LIBRARY

CIRCULATION SECTION

4500 N. WILSON ST.

LIBRARY LOAN COPY**DO NOT TRANSFER TO ANOTHER PERSON**

If you wish someone else to see this
report, send in name with report and
the library will change a loan.

LCN-7509 1-19-77

OPERATED BY
MARTIN MARIETTA ENERGY SYSTEMS, INC.
FOR THE UNITED STATES
DEPARTMENT OF ENERGY

Printed in the United States of America. Available from
National Technical Information Service
U.S. Department of Commerce
5285 Port Royal Road, Springfield, Virginia 22161
NTIS price codes--Printed Copy: A03; Microfiche A01

This report was prepared as an account of work sponsored by an agency of the United States Government. Neither the United States Government nor any agency thereof, nor any of their employees, makes any warranty, express or implied, or assumes any legal liability or responsibility for the accuracy, completeness, or usefulness of any information, apparatus, product, or process disclosed, or represents that its use would not infringe privately owned rights. Reference herein to any specific commercial product, process, or service by trade name, trademark, manufacturer, or otherwise, does not necessarily constitute or imply its endorsement, recommendation, or favoring by the United States Government or any agency thereof. The views and opinions of authors expressed herein do not necessarily state or reflect those of the United States Government or any agency thereof.

ORNL/TM-9693

Engineering Technology Division

**ALKACYCL,
A BASIC COMPUTER PROGRAM FOR THE ANALYSIS
OF ALKALI METAL RANKINE POWER CYCLES**

J. C. Moyers

Date Published - August 1985

Prepared by the
OAK RIDGE NATIONAL LABORATORY
Oak Ridge, Tennessee 37831
operated by
MARTIN MARIETTA ENERGY SYSTEMS, INC.
for the
U.S. Department of Energy
under Contract No. DE-AC05-84OR21400



3 4456 0054036 9

CONTENTS

ABSTRACT	1
1. INTRODUCTION	3
2. CYCLE DESCRIPTION	3
3. COMPONENT MODELING	5
3.1 TURBINE	5
3.2 MOISTURE SEPARATORS	5
3.3 REGENERATIVE FEED HEATERS	5
4. WORKING FLUID THERMODYNAMIC AND TRANSPORT PROPERTIES	6
5. PROGRAM STRUCTURE	6
REFERENCES	9
APPENDIX	11
Table A.1 Input Variables for ALKACYCL	13
ALKACYCL Program Listing	15
Sample Program Output	31

**ALKACYCL, A Basic Computer Program for the Analysis
of Alkali Metal Rankine Power Cycles**

J. C. Moyers

ABSTRACT

ALKACYCL is a computer program that analyzes Rankine power cycles utilizing an alkali metal as the cycle working fluid. Cycles may have from zero to three stages of regenerative feed heating. The program is written in BASICA language and can be used on an IBM-PC or PC-compatible computer with 128 kbytes of RAM. Output results include mass and energy balance information, cycle efficiency, and sizes and weights for piping and feed heaters. Listing and sample program output are included.

1. INTRODUCTION

Space power systems for applications requiring large amounts of electric power for long mission lifetimes probably will use nuclear reactors as the prime energy source and may utilize Rankine power conversion subsystems having an alkali metal as the cycle working fluid. Selection of the system temperature levels, top and bottom, is a critical step in the system design process, with the top temperature influencing the design and materials of construction of the reactor subsystem, the bottom temperature influencing the design and materials of construction of the heat rejection subsystem, and both temperatures influencing the power conversion subsystem cycle efficiency and, therefore, the sizing of both the reactor and heat rejection subsystems.

The computer program ALKACYCL was developed to provide a rapid means for performing mass and energy balances and determining the cycle efficiency of a Rankine power cycle operating between specified temperature levels and utilizing one of the alkali metals as the working fluid. Sizes and weights of interconnecting piping and of the regenerative feed heaters are estimated by the program also. (Future expansion of the program to include estimation of the weights of other components is planned.) Application of the program is limited to saturated vapor (either dry or of specified quality) entering the turbine. The program is written in BASICA language and is executable on an IBM-PC or PC-compatible computer having 128 kbytes of RAM.

2. CYCLE DESCRIPTION

An elementary flowsheet for the Rankine power cycle treated by ALKACYCL is shown in Fig. 1. Prime vapor flows to the main power turbine inlet, with a side stream flowing to the auxiliary turbine that drives the centrifugal feed pump. The power turbine exhausts to the condenser, from which condensate is scavenged by a jet pump that is driven by a side stream taken from the feed pump discharge. This jet pump provides adequate net positive suction head for the feed pump to prevent cavitation in the feed pump suction region. The bulk of the feed pump discharge stream progresses through a specified number of regenerative feed heaters (from zero to three heaters are allowed) to the boiler. The auxiliary turbine driving the feed pump exhausts to Heater 1 if at least one heater is specified; if no heaters are specified, the auxiliary turbine exhausts to the condenser.

A moisture separator may be inserted in the turbine following any stage to reduce the moisture content of the vapor flowing to the next stage. The removed liquid, along with some vapor that inevitably accompanies the liquid, is drained to the next-lower-temperature feed heater or, in the absence of a lower-temperature heater, to the condenser.

As noted previously, from zero to three regenerative feed heaters may be employed in the cycle. In proceeding from a zero-heater cycle to a three-heater cycle, the order of adding heaters is denoted by the heater numbers in Fig. 1. Heater 1 receives the exhaust vapor from the auxiliary turbine and drains either to Heater 3 if in a three-heater cycle or to the condenser if less than three heaters are used. Heater 2 receives vapor either extracted from an early stage of the power turbine, if the stage exit temperature is high enough to heat the boiler feed stream to the specified temperature level, or from the prime vapor stream. (Use of prime vapor for feed heating is poor practice, thermodynamically, but may be required if

boiler feed with little subcooling is required for stable boiler operation.) Heater 2 drains to Heater 1. Heater 3 receives vapor extracted from the lowest stage in the main turbine having an exit temperature high enough to cause the combined Heaters 1 and 3 to heat the feed stream leaving Heater 1 to its highest potential temperature (i.e., to the auxiliary turbine exhaust temperature minus the specified heater terminal temperature difference).

ORNL DWG 85-10324 R




Fig. 1 Rankine power cycle flowsheet.

3. COMPONENT MODELING

In the following subsections, the model assumed for each of the major cycle components is described briefly to inform the reader of simplifications built into the computer program and to provide a better understanding of the meaning of required items of program input data.

3.1 TURBINE

The power turbine has a specified number of stages, with an assumed temperature drop across each stage equal to the total cycle temperature difference divided by the number of stages. Each stage is assumed to have an aerodynamic efficiency equal to the input value for dry stage efficiency. The actual efficiency for the stage is then assumed to be the aerodynamic efficiency degraded by one percentage point per percent average stage moisture as determined from an isentropic expansion over the stage temperature range. Values for efficiency degradation due to moisture content assumed by various investigators in the literature range from 0.5 (Ref. 1) to 1.0 (Ref. 2) points per percent moisture at the exit of a turbine section. A value for turbine exhaust loss, due to the last stage leaving velocity, is specified in the input.

3.2 MOISTURE SEPARATORS

Either of two types of moisture separators may be inserted in the turbine expansion path following any turbine stage. An interstage separator provides means, within the turbine casing, for collecting and draining off accumulated moisture. An interspool separator is located external to the casing, with the working fluid stream exiting the casing, flowing through the separator, and reentering the casing prior to the following stage.

An interstage separator is assumed to remove 25% of the moisture present in the working fluid at the location of the separator.³ An associated penalty with this type of separator is the requirement for the removal of 0.25 lb of vapor with each pound of moisture removed.⁴

An interspool separator is assumed to remove 90% of the moisture in the working fluid as it enters the separator.⁴ Two performance penalties are associated with this type of separator. First, 0.1 lb of vapor accompanies each pound of removed moisture. Second, the working fluid undergoes a pressure drop of approximately two velocity heads (1.5 psi is assumed in ALKACYCL) as it passes through the separator. In addition, there is a weight penalty associated with an interspool separator. (In Ref. 4, an interspool separator was estimated to add 50 lb to the weight of a 450-kW potassium turbine.)

3.3 REGENERATIVE FEED HEATERS

Each of the feed heaters modeled in ALKACYCL consists of a shell-and-tube heat exchanger with a condensing section and a drain cooler section. The boiler feed stream flows on the shell side of the exchanger while the vapor-liquid mixture is condensed and subcooled inside the tubes. This arrangement is the reverse of normal utility power plant practice and is necessitated to assure management of the vapor-liquid interface in the zero-gravity environment.

Two items of input information are required from the program user to define feed heater performance. The terminal temperature difference, defined as the difference in temperature between the incoming vapor-liquid mixture and the leaving boiler feed stream, is specified. In addition, the drain cooler temperature difference, defined as the temperature difference between the drains leaving the heater and the incoming boiler feed stream, is required. The specified values of these items are common to all heaters.

4. WORKING FLUID THERMODYNAMIC AND TRANSPORT PROPERTIES

Thermodynamic and transport properties for potassium are provided by subroutines in ALKACYCL. Replacement subroutines are available for sodium properties, and others for cesium could be developed from available algorithms if the need arises.

The algorithms contained in the subroutine for potassium properties are from R. L. Graves,⁵ who also presented algorithms for the properties of cesium. The replacement subroutines for sodium are based on algorithms presented by G. H. Golden and J. V. Tokar.⁶

5. PROGRAM STRUCTURE

A listing of the program, a description of the required input variables, and a sample output are included in the Appendix. The subroutines included in the listing provide thermodynamic and transport properties for potassium. Following the main listing is a set of subroutines for sodium, which may be substituted in the program for cycle analyses using that working fluid. The functions of the various subroutines are as follows:

6000	-	returns saturated thermodynamic properties corresponding to the supplied temperature
6450	-	returns saturated liquid specific volume at the supplied temperature
6500	-	returns saturation temperature for the supplied pressure
7000	-	returns saturation temperature for the supplied saturated liquid enthalpy
7200	-	returns transport properties for the supplied saturation temperature

Input data are in Lines 7500-7760 and are defined in Table A.1 in the Appendix. In general, data in Lines 7500-7530 provide the information required for determination of the turbine expansion line. Data in Lines 7540-7550 define the remainder of the cycle. System capacity and desired degree of detail in the output are specified in Lines 7560-7570. If piping and feed heater sizing and weight estimates are not desired, no further input data are required. If this detail is desired, the following data must be provided. Lines 7590-7670 label the various interconnecting piping runs and provide their respective lengths, the physical state of the working fluid as it flows through the line, and the number of parallel pipes for each run (reflecting the modularity of the system). Lines 7690-7740 contain data setting design velocities for the piping and giving the physical properties of the materials of construction. Lines 7750-7760 contain the outside diameter and wall thickness of the tubing to be used in the feed heater bundles.

REFERENCES

1. J. H. Pitts and C. E. Walter, "Conceptual Design of a 10-MWe Nuclear Rankine System for Space Power," *Spacecraft* 7(3), 259-265 (March 1970).
2. A. J. Glassman, *Alkali-Metal Turbine Geometry as Affected by Blade-Speed Limitations*, NASA TN-2777, Natl. Aeronautics and Space Administration, April 1965.
3. G. Samuels et.al., *Design Study of a 200 MW(e) Alkali Metal/Steam Binary Power Plant Using Coal-Fired Fluidized Bed Furnace*, ORNL TM-6041, Union Carbide Corp. Nuclear Div., Oak Ridge Natl. Lab., April 1978.
4. Garrett Corp., *Turbine, Bearing, and Seal Parametric Design*, vol. 1 of *Potassium Turboalternator (KTA) Preliminary Design Study*, NASA CR-1498, Natl. Aeronautics and Space Administration, April 1970.
5. R. L. Graves, *Documentation of the Engineering Properties of Cesium and Potassium*, ORNL-5344, Union Carbide Corp. Nuclear Div., Oak Ridge Natl. Lab., November 1977.
6. G. H. Golden and J. V. Tokar, *Thermophysical Properties of Sodium*, ANL-7323, Argonne Natl. Lab., August 1967.

APPENDIX

Table A.1 Input variables for ALKACYCL

Variable	Location ^a	Definition
TBOIL	7500-1	Boiler outlet temperature, R
XBOIL	7500-2	Boiler outlet quality, fraction
TCON	7500-3	Turbine outlet temperature, R
NS%	7500-4	Number of turbine stages
GEFF	7500-5	Generator efficiency, fraction
DEFF	7500-6	Dry turbine efficiency (for all stages), fraction
EXLOSS	7500-7	Turbine exhaust loss, Btu/lb
SEP%	7520	Moisture separator index (set of NS% values) 0 = no separator 1 = interstage separator 2 = interspool separator
PTEFF	7540-1	Auxiliary turbine efficiency, fraction
PEFF	7540-2	Feed pump efficiency, fraction
NH%	7540-3	Number of feed heater stages
DELTPT	7540-4	Auxiliary turbine temperature difference, R
JETRAT	7540-5	Condenser scavenging jet pump flow ratio, suction/driving
DELPSYS	7540-6	Total system frictional pressure drop, psi
SCCON	7540-7	Condenser subcooling, R
TTD	7540-8	Feed heater terminal temperature difference, R
DCDT	7540-9	Drain cooler temperature difference, R
BFSC	7540-10	Boiler feed subcooling, R
KWOUT	7560-1	System electrical output, kW
DETAIL%	7560-2	Output detail index 0 = no piping or heaters sizing or weights 1 = full output

The following input data are not needed if DETAIL% = 0. If DETAIL% = 1, then 23 sets of piping run data (each set consisting of a value for LLBL\$, LG, TY, and NUM%) are required in Lines 7600-7670.

LLBL\$(n)	7600- 7670	Label for pipe run n (up to 16 characters)
LG(n)	7670	Length for pipe run n, ft
TY(n)	7670	Physical state of fluid flowing in pipe run n 1 = vapor 2 = wet mixture 3 = liquid
NUM%(n)	7670	Number of parallel pipes for pipe run n
NPTA%	7700-1	Number of temperature-strength data pairs for high-temperature alloy
NPTB%	7700-2	Number of temperature-strength data pairs for low-temperature alloy
TMAT	7700-3	Temperature above which high-temperature alloy is to be used
VELV	7700-4	Design velocity for vapor lines, ft/s
VELM	7700-5	Design velocity for wet mix. lines, ft/s

VELL	7700-6	Design velocity for liquid lines, ft/s
RHOA	7700-7	Density of high-temperature alloy, lb/in. ³
RHOB	7700-8	Density of low-temperature alloy, lb/in. ³
KA	7700-9	Thermal conductivity of high-temperature alloy, Btu/ft-h-F
KB	7700-10	Thermal conductivity of low-temperature alloy, Btu/ft-h-F
TSIG(m)	7720-1, 3,5...	Temperatures for which corresponding high-temperature alloy strengths are known, R (NPTA% values, input in odd-number locations in Line 7720)
SIGA(m)	7720-2, 4,6...	Design strength for high-temperature alloy corresponding to TSIG(m), psi (NPTA% values, in even-number locations in Line 7720)
SIGB(m)	7740-1, 2,3...	Design strength for low-temperature alloy corresponding to TSIG(m), psi (NPTB% values, corresponding to first NPTB% values for TSIG())
ODTUBE	7760-1	Outside diameter of tubing used in feed heaters, in.
WALLT	7760-2	Wall thickness of tubing used in feed heaters, in.

^a Location denotes line number and item location within that line (e.g., 7500-3 denotes the third data item in Line 7500).

LISTING OF PROGRAM ALKACYCL

```

5 'PROGRAM *ALKACYCL* developed by John C. Moyers, ORNL
10 WIDTH "LPT1:",120
20 LPRINT CHR$(15)
30 'THIS PROGRAM CALCULATES THE EXPANSION LINE STATE POINTS AND FLOW RATIOS
40 'FOR A MULTI-STAGE TURBINE WHICH MAY HAVE INTERSTAGE MOISTURE REMOVAL AND/
50 'OR INTERSPPOOL MOISTURE REMOVAL. A TURBINE CYCLE HEAT BALANCE IS ALSO
60 'CALCULATED, FOR 0 TO 3 HEATERS. DATA IS LOCATED IN STATEMENTS 7500-7800.
70 'THERMOPHYSICAL PROPERTIES OF WORKING FLUID ARE DEFINED IN SUBROUTINES
80 'LOCATED IN STATEMENTS 6000-7499.
90 CLOCK$=TIME$:DAY$=DATE$
100 DIM LLBL$(23),LG(23),TY(23),NUM%(23),TL(23),PL(23),MF(23),CFS(23),ID(23),WA
LL(23),WT(23),SIGMAL(23)
110 READ TB0IL,XBOIL,TCON,NS%,GEFF,DEFF,EXLOSS
120 FOR I%=1 TO NS%:READ SEP%(I%):NEXT I% '0=NO SEP;1=INTERSTAGE;2=INTERSPOOL
130 T(0)=TB0IL
140 DELTS=(T(0)-TCON)/NS% 'UNIFORM STAGE DELTA T
150 T=T(0)
160 GOSUB 6000
170 H(0)=HF+XBOIL*HFG:S(0)=SF+XBOIL*SFG:P(0)=14.696*P:X(0)=XBOIL
180 SVV(0)=VF+X(0)*(VG-VF):SVL(0)=VF
190 DELH(0)=0!:FLORAT(0)=1!
200 FOR N%=1 TO NS%
210 T(N%)=T(N%-1)-DELTS:T=T(N%)
220 GOSUB 6000
230 P(N%)=14.696*P:HF(N%)=HF
240 XS=(S(N%-1)-SF)/SFG:HS=HF+XS*HFG
250 EFF(N%)=DEFF-1!+(X(N%-1)+XS)/2! '1% PT LOSS IN EFF/1% AVG MOISTURE IN STAGE
260 H(N%)=H(N%-1)-(H(N%-1)-HS)*EFF(N%):HE(N%)=H(N%):XE(N%)=(H(N%)-HF)/HFG:SE(N%)
)=SF+XE(N%)*SFG:SVV(N%)=VF+XE(N%)*(VG-VF):SVL(N%)=VF 'ELEP PROP'S.
270 IF N%=NS% THEN H(N%)=H(N%)+EXLOSS 'UEEP ENTHALPY
280 X(N%)=(H(N%)-HF)/HFG 'UEEP QUALITY
290 IF N%=NS% THEN SVV(N%)=VF+X(N%)*(VG-VF)
300 S(N%)=SF+X(N%)*SFG
310 DELH(N%)=H(N%-1)-H(N%) 'STAGE ENTHALPY CHANGE
320 FLORAT(N%)=1! 'FLOW TO FOLLOWING STAGE/FLOW TO THIS STAGE
330 ON SEP%(N%)+1 GOTO 570,350,430
340 'INTERSTAGE MOISTURE REMOVAL
350 LSEP(N%)=.25*(1!-X(N%)) '25% OF STAGE EXIT MOISTURE REMOVED
360 VSEP(N%)=.25*LSEP(N%) '1/4 # VAPOR REMOVED PER # LIQUID REMOVED
370 X(N%)=(X(N%)-VSEP(N%))/(1!-VSEP(N%)-LSEP(N%)) 'NEW QUALITY TO NEXT STAGE
380 FLORAT(N%)=1!-VSEP(N%)-LSEP(N%) 'FLOW RATE TO NEXT STAGE/FLOW TO THIS STAGE
390 S(N%)=SF+X(N%)*SFG:SVV(N%)=VF+X(N%)*(VG-VF)
400 H(N%)=(H(N%)-LSEP(N%)*HF(N%)-VSEP(N%)*H(N%))/FLORAT(N%)
410 GOTO 570
420 'INTERSPOOL MOISTURE REMOVAL
430 LSEP(N%)=.9*(1!-X(N%)) '90% OF MOISTURE PRESENT IS REMOVED
440 VSEP(N%)=.1*LSEP(N%) '0.1# VAPOR REMOVED PER # LIQUID REMOVED
450 TEXIT(N%)=T(N%)
460 FLORAT(N%)=1!-VSEP(N%)-LSEP(N%)

```

```

470 H(N%)=(H(N%)-LSEP(N%)*HF(N%)-VSEP(N%)*H(N%))/FLORAT(N%)
480 P(N%)=P(N%)-1.5
490 P=P(N%)/14.696
500 GOSUB 6500
510 T(N%)=TEMP
520 T=T(N%)
530 GOSUB 6000
540 X(N%)=(H(N%)-HF)/HFG
550 S(N%)=SF+X(N%)*SFG:SVV(N%)=VF+X(N%)*(VB-VF):SVL(N%)=VF
560 DELTS=(T(N%)-TCOM)/(NS%-N%)
570 NEXT N%
580 LPRINT "TURBINE EXPANSION LINE CALCULATION";TAB(74);CLOCK$;" ";DAY$:LPRINT
590 LPRINT "THROT. T, R = ";TBOIL;" COND. T, R = ";TCOM
600 LPRINT "NO. OF STAGES = ";NS%;" DRY STAGE EFF, % = ";100*DEFF
610 LPRINT "EXHAUST LOSS, BTU/LBM = ";EXLOSS:LPRINT
620 LPRINT "STAGE";TAB(12);"EXIT T";TAB(26);"EXIT P";TAB(39);"EXIT S";TAB(49);"
EXIT H";TAB(63);"EXIT X";TAB(74);"FLOW RATIO";TAB(86);"DELTA H":LPRINT
630 FOR N%=0 TO NS%
640 LPRINT N%;TAB(12);
650 LPRINT USING "####.#      ";T(N%);
660 LPRINT USING "####.##### ";P(N%),S(N%),H(N%),X(N%),FLORAT(N%),DELH(N%)
670 NEXT N%
680 FOR N%=1 TO NS%
690 IF SEP%(N%)>0 GOTO 720
700 NEXT N%
710 GOTO 780
720 LPRINT:LPRINT:LPRINT "MOISTURE SEPARATIONS (LIQ. OUT AND VAP. OUT ARE PER
LBM ENTERING STAGE)":LPRINT
730 LPRINT "STAGE";TAB(14);TYPE;TAB(25);LIQ. OUT;TAB(36);LIQ. H;TAB(49);"
VAP. OUT";TAB(60);VAP. H;TAB(73);VAP. S;TAB(86);ELEP X":LPRINT
740 FOR N%=1 TO NS%
750 IF SEP%(N%)=1 THEN LPRINT N%;TAB(12);INTERSTAGE":LPRINT USING "####.#####
";LSEP(N%),HF(N%),VSEP(N%),HE(N%),SE(N%),XE(N%)
760 IF SEP%(N%)=2 THEN LPRINT N%;TAB(12);INTERSPOOL":LPRINT USING "####.#####
";LSEP(N%),HF(N%),VSEP(N%),HE(N%),SE(N%),XE(N%)
770 NEXT N%
780 LPRINT:LPRINT:LPRINT "      LAST STAGE ELEP: H = ";HE(NS%);" S = ";SE(NS%)
;" X = ";XE(NS%)
790 READ PTEFF,PEFF,NH%,DELTPT,JETRAT,DELPSYS,SCCON,TTD,DCDT,BFSC
800 READ KWOUT,DETAILX
810 LPRINT CHR$(12):LPRINT "TURBINE CYCLE CALCULATION (PER LB PRIME VAPOR)":TAB
(74);CLOCK$;" ";DAY$
820 LPRINT:LPRINT "NO. OF HEATER STAGES = ";NH%;"      PUMP TURB. DELTA T = ";D
ELTPT
830 LPRINT "JET PUMP FLOW RATIO = ";JETRAT;"      PUMP TURB. EFF. = ";PTEFF;
PUMP EFF. = ";PEFF;"      GEN. EFF. = ";GEFF
840 LPRINT "SYSTEM PRESSURE DROP = ";DELPSYS;"      CONDENSER SUBCOOLING = ";SCCO
N
850 LPRINT "HEATER TERMINAL TEMP. DIFF. = ";TTD;"      DRAIN COOLER DELTA T = ";D
CDT

```

```

860 LPRINT "BOILER FEED SUBCOOLING = ";BFSC:LPRINT
870 TOUT=T(0)-DELTPT:TEXT(1)=TOUT:T=TOUT
880 GOSUB 6000
890 PTOUT=14.696*P:XS=(S(0)-SF)/SFG:HS=HF+XS*HFG:HHF(1)=HF
900 HOUT=H(0)-(H(0)-HS)*PTEFF:XOUT=(HOUT-HF)/HFG:SOUT=SF+XOUT*SFG:SVVOUT=VF+XOU
T*(VG-VF):SVLOUT=VF 'PUMP TURBINE ELEP PROPERTIES
910 T=TCON-SCCON:THW=T
920 GOSUB 6000
930 VFHW=VF 'HOTWELL LIQUID VF
940 HHW=HF 'HOTWELL ENTHALPY
950 WORKP=(JETRAT+1)/JETRAT*(P(0)-P(NS%)+DELPSYS)*144!*VFHW 'PUMP WORK,FT.LBS/L
BM BOILER VAPOR
960 WORKT=WORKP/(PEFF*778) 'TURBINE WORK, BTU/LBM BOILER VAPOR
970 HPUMP=HHW+WORKT:H=HPUMP:T=THW
980 GOSUB 7000
990 TPUMP=TEMP
1000 FLOWPT=WORKT/(H(0)-HOUT) 'PUMP TURBINE FLOW, LBM/LBM BOILER VAPOR
1010 FLOW(0)=1!-FLOWPT:WORK=0!:QSEP(0)=0!:QSEP(1)=0!:FLOSEP(0)=0!:FLOSEP(1)=0!
1020 '**** ZERO OR ONE FEED HEATER
1030 FOR IX=1 TO NS%
1040 FLOW(IX)=FLOW(IX-1)*FLORAT(IX-1) 'STAGE FLOW, LBM/LBM PRIME VAPOR
1050 WORKS(IX)=FLOW(IX)*DELH(IX) 'STAGE WORK, BTU/LBM PRIME VAPOR
1060 WORK=WORK+WORKS(IX)
1070 IF SEP%(IX)=0 GOTO 1140
1080 IF T(IX)<TOUT GOTO 1120
1090 QSEP(1)=QSEP(1)+FLOW(IX)*(LSEP(IX)*HF(IX)+VSEP(IX)*HE(IX))
1100 FLOSEP(1)=FLOSEP(1)+FLOW(IX)*(LSEP(IX)+VSEP(IX))
1110 GOTO 1140
1120 QSEP(0)=QSEP(0)+FLOW(IX)*(LSEP(IX)*HF(IX)+VSEP(IX)*HE(IX))
1130 FLOSEP(0)=FLOSEP(0)+FLOW(IX)*(LSEP(IX)+VSEP(IX))
1140 NEXT IX
1150 ON NH%+1 GOTO 1160,1220,1380,1380
1160 QADD=H(0)-HPUMP 'BOILER HEAT ADDED/LBM PRIME VAPOR
1170 QREJ=FLOWPT*HOUT+FLOW(NS%)*H(NS%)+QSEP(0)+QSEP(1)-HHW
1180 CYCEFF=WORK/QADD 'DOES NOT INCLUDE GENERATOR LOSSES
1190 TFW=TPUMP:SVLFW=VFHW
1200 LPRINT "*** NO FEED HEATERS ***":LPRINT
1210 ON DETAIL%+1 GOTO 4500,2660
1220 LPRINT "*** ONE FEED HEATER USING PUMP TURBINE EXHAUST ***":LPRINT
1230 TTO(1)=TPUMP:HTO(1)=HPUMP:TDRAIN(1)=TPUMP+DCDT:T=TDRAIN(1)
1240 GOSUB 6000
1250 HDRAIN(1)=HF:SVLDRN1=VF
1260 HFROM(1)=HTO(1)+FLOWPT*HOUT+QSEP(1)-HDRAIN(1)*(FLOSEP(1)+FLOWPT)
1270 T=TTO(1):H=HFROM(1)
1280 GOSUB 7000
1290 TFROM(1)=TEMP:T=TEMP
1300 GOSUB 6450
1310 SVLFRM1=VF
1320 SVLT01=VFHW
1330 QADD=H(0)-HFROM(1)

```

```

1340 QREJ=FLOW(NS%)*H(NS%)+(FLOWPT+FLOSEP(1))*HDRAIN(1)+QSEP(0)-HHW
1350 CYCEFF=WORK/QADD
1360 TFW=TFROM(1):SVLFW=SVLFRM1
1370 ON DETAIL%+1 GOTO 4500,2660
1380 *****TWO FEED HEATERS*****
1390 TR=(T(0)-BFSC+TCON)/2
1400 GOSUB 7200  'TO OBTAIN SPECIFIC HEAT (CP) OF LIQUID AT AN AVG. TEMP.
1410 TX=T(0)-BFSC+TTD
1420 FOR I%=0 TO NS%
1430 IF T(I%)<TX GOTO 1450
1440 NEXT I%
1450 TEXT(2)=T(I%-1):NSEXT2%=I%-1:HEXT(2)=H(I%-1)
1460 TFRDM(2)=T(0)-BFSC:T=TFRDM(2)
1470 GOSUB 6000
1480 HFROM(2)=HF:SVLFRM2=VF:SVLTO1=VFHW:SVLFW=VF
1490 TDRAIN(1)=TPUMP+DCDT:T=TDRAIN(1)
1500 GOSUB 6000
1510 HDRAIN(1)=HF:TTD(1)=TPUMP:HTD(1)=HPUMP:SVLDRN1=VF
1520 IF NH%>3 GOTO 1900
1530 FLOEXT(2)=(HFROM(2)-HPUMP-QSEP(1)-FLOWPT*HOUT+HDRAIN(1)*(FLOSEP(1)+FLOWPT)
)/(HEXT(2)-QSEP(1)/FLOW(0)+HDRAIN(1)*(FLOSEP(1)/FLOW(0)-1))
1540 IF FLOEXT(2)<=0! GOTO 2640
1550 HTD(2)=(HFROM(2)-FLOEXT(2)*(HEXT(2)-CP*DCDT))/(1-FLOEXT(2)):HFRDM(1)=HTD(2)
)
1560 H=HTD(2):T=TOUT
1570 GOSUB 7000
1580 TTD(2)=TEMP:TFRDM(1)=TEMP:T=TEMP:GOSUB 6450
1590 SVLFRM1=VF
1600 HDRAIN(2)=HFROM(1)+CP*DCDT:H=HDRAIN(2):T=TOUT
1610 GOSUB 7000
1620 TDRAIN(2)=TEMP:T=TEMP:GOSUB 6450
1630 SVLDRN2=VF
1640 QSEP(1)=QSEP(1)*(1!-FLOEXT(2)/FLOW(0))
1650 FLOSEP(1)=FLOSEP(1)*(1!-FLOEXT(2)/FLOW(0))
1660 QSEP(0)=QSEP(0)*(1!-FLOEXT(2)/FLOW(0))
1670 FLOSEP(0)=FLOSEP(0)*(1!-FLOEXT(2)/FLOW(0))
1680 IF NSEXT2%>0 GOTO 1760
1690 FLOW(0)=FLOW(0)-FLOEXT(2):WORK=0!
1700 FOR I%=1 TO NS%
1710 FLOW(I%)=FLOW(I%-1)*FLORAT(I%-1)
1720 WORKS(I%)=FLOW(I%)*DELH(I%)
1730 WORK=WORK+WORKS(I%)
1740 NEXT I%
1750 GOTO 1830
1760 WORK=0!
1770 FOR I%=1 TO NS%
1780 FLOW(I%)=FLOW(I%-1)*FLORAT(I%-1)
1790 IF I%=NSEXT2%+1 THEN FLOW(I%)=FLOW(I%)-FLOEXT(2)
1800 WORKS(I%)=FLOW(I%)*DELH(I%)
1810 WORK=WORK+WORKS(I%)

```

```

1820 NEXT I%
1830 QADD=H(0)-HFROM(2)
1840 QREJ=FLOW(NS%)*H(NS%)+(FLOEXT(2)+FLOWPT+FLOSEP(1))*HDRAIN(1)+QSEP(0)-HHW
1850 CYCEFF=WORK/QADD
1860 TFW=TFROM(2):SVLFW=SVLFRM2
1870 LPRINT "*** TWO FEED HEATERS FED FROM STAGE ";NSEXT2%;" AND FROM PUMP TURB
INE EXHAUST ***":LPRINT
1880 ON DETAIL%+1 GOTO 4500,2660
1900 TFROM(1)=TOUT-TTD:T=TFROM(1):TTD(2)=T
1910 GOSUB 6000
1920 HFROM(1)=HF:HTO(2)=HF:SVLFRM1=VF:SVLT02=VF
1930 TDRAIN(2)=TTD(2)+DCDT:T=TDRAIN(2)
1940 GOSUB 6000
1950 HDRAIN(2)=HF:SVLDRN2=VF
1960 FLOEXT(2)=(HFROM(2)-HTO(2))/(HEXT(2)-HDRAIN(2))
1970 QSEP1=QSEP(1)
1980 QSEP(1)=QSEP(1)*(1!-FLOEXT(2)/FLOW(0))
1990 FLOSEP1=FLOSEP(1)
2000 FLOSEP(1)=FLOSEP(1)*(1!-FLOEXT(2)/FLOW(0))
2010 HTO(1)=(HFROM(1)-FLOWPT*HOUT-QSEP(1)-FLOEXT(2)*HDRAIN(2)+(FLOWPT+FLOSEP(1)
+FLOEXT(2))*CP*DCDT)/(1-(FLOWPT+FLOSEP(1)+FLOEXT(2)))
2020 HDRAIN(1)=HTO(1)+CP*DCDT
2030 H=HTO(1):T=TOUT-200!
2040 GOSUB 7000
2050 TTD(1)=TEMP:TFROM(3)=TEMP
2060 TX=TFROM(3)+TTD
2070 FOR I%=1 TO NS%
2080 IF T(I%)>TX GOTO 2110
2090 TEXT(3)=T(I%-1):NSEXT3%=I%-1:HEXT(3)=H(I%-1)
2100 GOTO 2120
2110 NEXT I%
2120 IF NSEXT3%=NS% GOTO 2620
2130 HFROM(3)=HTO(1)
2160 T=TFROM(3):GOSUB 6000:SVLFRM3=VF:SVLT01=VF
2170 TDRAIN(1)=TTD(1)+DCDT:T=TDRAIN(1):GOSUB 6000
2190 SVLDRN1=VF
2320 QSEP(0)=0!:FLOSEP(0)=0!:QSEP(3)=0!:FLOSEP(3)=0!:FLOW(0)=1!-FLOWPT:WORK=0!
2330 TDRAIN(3)=TPUMP+DCDT:T=TDRAIN(3)
2340 GOSUB 6000
2350 HDRAIN(3)=HF:SVLDRN3=VF
2360 FOR I%=1 TO NSEXT3%
2370 FLOW(I%)=FLOW(I%-1)*FLORAT(I%-1)
2380 IF I%=NSEXT2%+1 THEN FLOW(I%)=FLOW(I%)-FLOEXT(2)
2390 WORKS(I%)=FLOW(I%)*DELH(I%)
2400 WORK=WORK+WORKS(I%)
2410 IF T(I%)>=TOUT GOTO 2440
2420 QSEP(3)=QSEP(3)+FLOW(I%)*(LSEP(I%)*HF(I%)+VSEP(I%)*HE(I%))
2430 FLOSEP(3)=FLOSEP(3)+FLOW(I%)*(LSEP(I%)+VSEP(I%))
2440 NEXT I%
2450 HTO(3)=HPUMP

```

```

2460 FLOEXT(3)=(HFROM(3)-HTD(3)-QSEP(3)+FLOSEP(3)*HDRAIN(3)-(FLOWPT+FLOEXT(2)+F
LOSEP(1))*(HDRAIN(1)-HDRAIN(3)))/(HEXT(3)-HDRAIN(3))
2470 IF FLOEXT(3)<=0! GOTO 2620
2480 FOR I%=NSEXT3%+1 TO NS%
2490 FLOW(I%)=FLOW(I%-1)*FLORAT(I%-1)
2500 IF I%=NSEXT3%+1 THEN FLOW(I%)=FLOW(I%)-FLOEXT(3)
2510 WORKS(I%)=FLOW(I%)*DElh(I%)
2520 WORK=WORK+WORKS(I%)
2530 QSEP(0)=QSEP(0)+FLOW(I%)*(LSEP(I%)*HF(I%)+VSEP(I%)*HE(I%))
2540 FLOSEP(0)=FLOSEP(0)+FLOW(I%)*(LSEP(I%)+VSEP(I%))
2550 NEXT I%
2560 QADD=H(0)-HFROM(2)
2570 QREJ=FLOW(NS%)*H(NS%)+(FLOEXT(2)+FLOWPT+FLOSEP(1)+FLOEXT(3)+FLOSEP(3))*HDR
AIN(3)+QSEP(0)-HHW
2580 CYCEFF=WORK/QADD
2590 TFW=TFROM(2)
2600 LPRINT "*** THREE FEED HEATERS FED FROM STAGES ";NSEXT2%;" AND ";NSEXT3%;"
AND FROM PUMP TURBINE EXHAUST ***":LPRINT
2610 ON DETAIL%+1 GOTO 4500,2660
2620 LPRINT "*** NO ROOM FOR HEATER 3 -- RUN ABORTED ***"
2630 GOTO 8000
2640 LPRINT "*** NO ROOM FOR HEATER 2 -- RUN ABORTED ***"
2650 GOTO 8000
2660 FACTOR=KWOUT*3412/(WORK*GEFF*3600) 'LB/SEC
2670 MMAIN=FACTOR
2680 MFLOPT=FACTOR*FLOWPT
2690 FOR I%=1 TO NS%:MFLO(I%)=FLOW(I%)*FACTOR:MLSEP(I%)=MFLO(I%)*LSEP(I%):MVSEP
(I%)=MFLO(I%)*VSEP(I%):NEXT I%
2700 MQADD=QADD*FACTOR*3600/3412 'KW(T)
2710 MQREJ=QREJ*FACTOR*3600/3412 'KW(T)
2720 MFLOEX2=FLOEXT(2)*FACTOR:MFLOEX3=FLOEXT(3)*FACTOR
2730 MFLOSE0=FLOSEP(0)*FACTOR:MFLOSE1=FLOSEP(1)*FACTOR:MFLOSE3=FLOSEP(3)*FACTOR
2740 MQSEP0=QSEP(0)*FACTOR:MQSEP1=QSEP(1)*FACTOR:MQSEP3=QSEP(3)*FACTOR
2750 MFLODR(2)=MFLOEX2
2760 IF NH%>0 THEN MFLODR(1)=MFLOEX2+MFLOPT+MFLOSE1
2770 IF NH%>3 THEN MFLODR(3)=MFLODR(1)+MFLOSE3+MFLOEX3:HINT03=(MFLODR(1)*HDRAIN
(1)+MFLOEX3*HEXT(3)+MQSEP3)/MFLODR(3):I=3:TTO(3)=TPUMP ELSE I=1
2780 MFLOFP=MMAIN*(JETRAT+1)/JETRAT
2790 HINT02=HEXT(2)
2800 IF NH%>0 THEN HINT01=(MFLOEX2*HDRAIN(2)+MFLOPT*HOUT+MQSEP1)/MFLODR(1)
2810 IF NH%>0 THEN HTCON=(MFLODR(I)*HDRAIN(I)+MFLO(NS%)*H(NS%)+MQSEP0)/MMAIN E
lse HTCON=(MFLO(NS%)*H(NS%)+MQSEP0)/MMAIN
2820 IF NH%>0 GOTO 2970
2830 T=TOUT
2840 GOSUB 6000
2850 XINT01=(HINT01-HF)/HFG
2860 VINT01=MFLODR(1)*(VF+XINT01*(VG-VF))
2870 IF NH%<2 THEN 2970
2880 T=TEXT(2)
2890 GOSUB 6000

```

```

2900 XINT02=(HINT02-HF)/HFG
2910 VINT02=MFL0EX2*(VF+XINT02*(VG-VF))
2920 IF NH%<3 THEN 2970
2930 T=TEXT(3)
2940 GOSUB 6000
2950 XINT03=(HINT03-HF)/HFG
2960 VINT03=MFL0DR(3)*(VF+XINT03*(VG-VF))
2970 T=TCON
2980 GOSUB 6000
2990 XTOCON=(HTOCON-HF)/HFG
3000 VTOCON=MMAIN*(VF+XTOCON*(VG-VF))
3010 IF NH%=0 GOTO 3100
3020 HHF(2)=HF(NSEXT2%):HHF(3)=HF(NSEXT3%)
3030 HMIX(1)=HINT01:HMIX(2)=HINT02:HMIX(3)=HINT03
3040 FOR I%=1 TO NH%
3050 QHX(I%)=MMAIN*(HFRDM(I%)-HTO(I%))
3060 TZ=TTO(I%)+(TFRDM(I%)-TTO(I%))*((HHF(I%)-HDRAIN(I%))/(HMIX(I%)-HDRAIN(I%)))
3070 LMTDC(I%)=(TZ-TFRDM(I%))/LOG((TEXT(I%)-TFRDM(I%))/(TEXT(I%)-TZ))
3080 LMTDD(I%)=((TEXT(I%)-TZ)-(TDRAIN(I%)-TTO(I%)))/LOG((TEXT(I%)-TZ)/(TDRAIN(I%)-TTO(I%)))
3090 NEXT I%
3100 '**** PIPING DESIGN ****'
3110 FOR I%=1 TO 23:READ LLBL$(I%),LG(I%),TY(I%),NUM%(I%):NEXT I%
3120 READ NPTA%,NPTB%,TMAT,VELV,VELM,VELL,RHOA,RHOB,KA,KB
3130 FOR I%=1 TO NPTA%:READ TSIG(I%),SIBA(I%):NEXT I%
3140 FOR I%=1 TO NPTB%:READ SIGB(I%):NEXT I%
3150 TL(1)=T(0):PL(1)=P(0):MF(1)=MMAIN:CFS(1)=MMAIN*SVV(0)
3160 TL(2)=T(0):PL(2)=P(0):MF(2)=MFL0(1):CFS(2)=MF(2)*SVV(0)
3170 TL(3)=TCON:PL(3)=P(NS%):MF(3)=MFL0(NS%):CFS(3)=MF(3)*SVV(NS%)
3180 TL(4)=TCON:PL(4)=P(NS%):MF(4)=MMAIN:CFS(4)=VTOCON
3190 TL(5)=THW:PL(5)=P(NS%):MF(5)=MMAIN:CFS(5)=MMAIN*VFHW
3200 TL(6)=THW:PL(6)=PL(5)+10!:MF(6)=MFL0FP:CFS(6)=MF(6)*VFHW
3210 TL(7)=TTO(3):PL(7)=P(0)+DELPYS:MF(7)=MF(5):CFS(7)=MF(7)*VFHW
3220 TL(8)=TTO(1):PL(8)=PL(7):MF(8)=MF(5):CFS(8)=MF(8)*SVLTO1
3230 TL(9)=TTO(2):PL(9)=PL(7):MF(9)=MF(5):CFS(9)=MF(9)*SVLFRM1
3240 TL(10)=TFW:PL(10)=PL(7):MF(10)=MF(5):CFS(10)=MF(10)*SVLFW
3250 TL(11)=T(0):PL(11)=P(0):MF(11)=MFL0PT:CFS(11)=MF(11)*SVV(0)
3260 TL(12)=TOUT:PL(12)=PTOUT:MF(12)=MFL0PT:CFS(12)=MF(12)*SVVDUT
3270 TL(13)=TEXT(2):PL(13)=P(NSEXT2%):MF(13)=MFL0EX2:CFS(13)=MF(13)*SVV(NSEXT2%)
3280 TL(14)=TDRAIN(2):PL(14)=PL(13):MF(14)=MFL0DR(2):CFS(14)=MF(14)*SVLDRN2
3290 FLAG1%=0:FLAG2%=0
3300 FOR I%=1 TO NS%
3310 IF SEP%(I%)=0 GOTO 3370
3320 IF FLAG1%=1 GOTO 3340
3330 IF T(I%)>=TOUT THEN TL(15)=T(I%):PL(15)=P(I%):CFS(15)=MFL0SE1*(MVSEP(I%)*$VV(I%)+MLSEP(I%)*SVL(I%))/(MVSEP(I%)+MLSEP(I%)):FLAG1%=1:GOTO 3370 ELSE GOTO 3340
3340 IF FLAG2%=1 GOTO 3360

```

```

3350 IF I%<=NSEXT3% THEN TL(19)=T(I%):PL(19)=P(I%):CFS(19)=MFLOSE3*(MVSEP(I%)*$VV(I%)+MLSEP(I%)*SVL(I%))/(MVSEP(I%)+MLSEP(I%)):FLAG2%=1:GOTO 3370 ELSE GOTO 3360
3360 TL(22)=T(I%):PL(22)=P(I%):CFS(22)=MFLOSE0*(MVSEP(I%)*$VV(I%)+MLSEP(I%)*SVL(I%))/(MVSEP(I%)+MLSEP(I%)):GOTO 3380
3370 NEXT I%
3380 MF(15)=MFLOSE1:MF(19)=MFLOSE3:MF(22)=MFLOSE0
3390 TL(16)=TDUT:PL(16)=PTOUT:MF(16)=MFLODR(1):CFS(16)=VINT01
3400 TL(17)=TDRAIN(1):PL(17)=PTOUT:MF(17)=MFLODR(1):CFS(17)=MF(17)*SVLDRN1
3410 TL(18)=TEXT(3):PL(18)=P(NSEXT3%):MF(18)=MFLOEX3:CFS(18)=MFLOEX3*$VV(NSEXT3%)
3420 TL(20)=TEXT(3):PL(20)=PL(18):MF(20)=MFLODR(3):CFS(20)=VINT03
3430 TL(21)=TDRAIN(3):PL(21)=PL(20):MF(21)=MF(20):CFS(21)=MF(21)*SVL(NSEXT3%)
3440 TL(23)=TPUMP:PL(23)=PL(7):MF(23)=MF(1)/JETRAT:CFS(23)=MF(23)*VFHW
3450 DRVOL(1)=CFS(17):DRVOL(2)=CFS(14):DRVOL(3)=CFS(21)
3460 TOTWT=0!
3470 FOR I%=1 TO 23
3480 ON TY(I%) GOTO 3490,3500,3510
3490 VELO=VELV:GOTO 3530
3500 VELO=VELM:GOTO 3530
3510 VELO=VELL
3520 IF NUM%(I%)=0 GOTO 3830
3530 ID(I%)=13.54*SQR(CFS(I%)/(VELO*NUM%(I%)))
3540 IF TL(I%)>TMAT GOTO 3670
3550 RHO=RHOB
3560 SIGMA=0!
3570 FOR J%=1 TO NPTB%
3580 A=SIGB(J%)
3590 FOR K%=1 TO NPTB%
3600 IF K%=J% GOTO 3620
3610 A=A*(TL(I%)-TSIG(K%))/(TSIG(J%)-TSIG(K%))
3620 NEXT K%
3630 SIGMA=SIGMA+A
3640 NEXT J%
3650 SIGMAL(I%)=SIGMA
3660 GOTO 3780
3670 RHO=RHOA
3680 SIGMA=0
3690 FOR J%=1 TO NPTA%
3700 A=SIGA(J%)
3710 FOR K%=1 TO NPTA%
3720 IF K%=J% THEN 3740
3730 A=A*(TL(I%)-TSIG(K%))/(TSIG(J%)-TSIG(K%))
3740 NEXT K%
3750 SIGMA=SIGMA+A
3760 NEXT J%
3770 SIGMAL(I%)=SIGMA
3780 WALL(I%)=PL(I%)*ID(I%)/2!/SIGMA
3790 IF WALL(I%)<.02 THEN WALL(I%)=.02
3800 WT(I%)=37.7*NUM%(I%)*LG(I%)*RHO*WALL(I%)*(ID(I%)+WALL(I%))

```

```

3810 IF ID(I%)=0! THEN WT(I%)=0!
3820 TOTWT=TOTWT+WT(I%)
3830 NEXT I%
3840 **** FEED HEATER DESIGN ***
3850 READ ODTUBE,WALLT
3860 VINTO(1)=VINTO1:VINTO(2)=VINTO2:VINTO(3)=VINTO3
3870 PMIX(1)=PL(16):PMIX(2)=PL(13):PMIX(3)=PL(20)
3880 FINTO(1)=CFS(8):FINTO(2)=CFS(9):FINTO(3)=CFS(7)
3890 TMIX(1)=TL(16):TMIX(2)=TL(13):TMIX(3)=TL(20)
3900 SIGMAH(1)=SIGMAL(16):SIGMAH(2)=SIGMAL(13):SIGMAH(3)=SIGMAL(20)
3910 ON NH% GOTO 3960,3940,3920
3920 TLIQ(1)=(TL(8)+TL(9))/2!:TLIQ(2)=(TL(9)+TL(10))/2!:TLIQ(3)=(TL(7)+TL(8))/2!
3930 GOTO 3970
3940 TLIQ(1)=(TL(8)+TL(9))/2!:TLIQ(2)=(TL(9)+TL(10))/2!
3950 GOTO 3970
3960 TLIQ(1)=(TL(8)+TL(10))/2!
3970 IDTUBE=(ODTUBE-2*WALLT)/12:ODTUBE=ODTUBE/12
3980 FOR I%=1 TO 3
3990 IF VINTO(I%)<.0001 GOTO 4270
4000 NTUBES=VINTO(I%)/(NUM%(2)*VELV*.7854*IDTUBE^2)
4010 NTUBES(I%)=CINT(NTUBES)
4020 AREASH=FINTO(I%)/(NUM%(2)*NTUBES(I%)*VELL)
4030 PITCH(I%)=SQR(1.155*AREASH+.9069*ODTUBE^2)
4040 DHYD=1.1027*PITCH(I%)^2/ODTUBE-ODTUBE
4050 DSHELL(I%)=PITCH(I%)*SQR(4*NTUBES(I%)/3-1/3)
4060 VSHELL=FINTO(I%)/(NUM%(2)*.7854*(DSHELL(I%)^2-NTUBES(I%)*ODTUBE^2))
4070 TR=TLIQ(I%)
4080 GOSUB 7200
4090 HO(I%)=(6.7+.0041*(DHYD*VSHELL*3600*RHOFL*CP/K)^.793*EXP(41.8*CP*MU/K))*K/
DHYD
4100 TR=(TEXT(I%)+TDRAIN(I%))/2
4110 GOSUB 7200
4120 V=DRVOL(I%)*3600/(NUM%(2)*NTUBES(I%)*.7854*IDTUBE^2) 'FT/HR IN TUBES
4130 HI(I%)=.625*(K/IDTUBE)*(IDTUBE*V*RHOFL*CP/K)^.4
4140 IF TMIX(I%)>TMAT THEN KTUBE=KA ELSE KTUBE=KB
4150 UIC(I%)=1/(IDTUBE*LOG(ODTUBE/IDTUBE)/2/KTUBE+IDTUBE/HO(I%)/ODTUBE)
4160 UID(I%)=1/(1/HI(I%)+IDTUBE*LOG(ODTUBE/IDTUBE)/2/KTUBE+IDTUBE/HO(I%)/ODTUBE)
4170 QHXC(I%)=QHX(I%)*(1-((HHF(I%)-HDRAIN(I%))/(HMIX(I%)-HDRAIN(I%)))) 'BTU/SEC
4180 QHxD(I%)=QHX(I%)-QHXC(I%) 'BTU/SEC FOR DRAIN COOLER
4190 LENGTHC=QHXC(I%)*3600/(NUM%(2)*3.1416*IDTUBE*NTUBES(I%)*UIC(I%)*LMTDC(I%))
4200 LENGTHD=QHxD(I%)*3600/(NUM%(2)*3.1416*IDTUBE*NTUBES(I%)*UID(I%)*LMTDD(I%))
4210 LENGTH(I%)=LENGTHC+LENGTHD
4220 WALLSH(I%)=PMIX(I%)*DSHELL(I%)/2!/SIGMAH(I%)
4230 IF WALLSH(I%)<.001667 THEN WALLSH(I%)=.001667
4240 IF TMIX(I%)>TMAT THEN RHO=RHOA ELSE RHO=RHOB
4250 WTHTR(I%)=1.1*RHO*1728!*LENGTH(I%)*.7854*((DSHELL(I%)+2*WALLSH(I%))^2-DSHELL(I%)^2+NTUBES(I%)*(ODTUBE^2-IDTUBE^2)) 'INCLUDES 10% FOR CAPS AND TUBESHEETS

```

```

4260 TOTHTR=TOTHTR+WTHTR(I%)*NUM%(2)
4270 NEXT I%
4500 LPRINT "STAGE";TAB(12);"STAGE FLOW";TAB(26);"STAGE WORK";TAB(39);"LIQ. SEP"
";TAB(51);"VAP. SEP";TAB(63);"STAGE EFF.":LPRINT
4510 FOR I%=1 TO NS%
4520 LPRINT I%;TAB(12);
4530 LPRINT USING "#####.####";FLOW(I%),WORKS(I%),FLOW(I%)*LSEP(I%),FLOW(I%)*
VSEP(I%),EFF(I%)
4540 NEXT I%
4550 LPRINT:LPRINT "HEAT ADDED = ";QADD;" HEAT REJECTED = ";QREJ
4560 LPRINT "TOTAL TURBINE WORK = ";WORK;" CYCLE EFFICIENCY = ";CYCEFF;""
PLANT EFF. =";CYCEFF*GEFF:LPRINT
4570 LPRINT "PUMP TURBINE FLOW = ";FLOWPT;" PUMP TURBINE EXH. ENTHALPY = ";H
OUT
4580 LPRINT "PUMP TURBINE WORK = ";WORKT;" BOILER FEED TEMPERATURE = ";TFW;""
DEG. R"
4590 LPRINT "EXTRACTION FLOW TO HEATER 2 = ";FLOEXT(2);" EXTRACTION FLOW TO
HEATER 3 = ";FLOEXT(3)
4600 LPRINT:LPRINT
4610 LPRINT "TEXT(2) = ";TEXT(2);TAB(25);"TOUT = ";TOUT;TAB(50);"TEXT(3) = ";TEXT(3)
4620 LPRINT "HEXT(2) = ";HEXT(2);TAB(25);"HOUT = ";HOUT;TAB(50);"HEXT(3) = ";HEXT(3)
4630 LPRINT "FLOEXT(2) = ";FLOEXT(2);TAB(25);"FLOWPT = ";FLOWPT;TAB(50);"FLOEXT(3) = "
;FLOEXT(3)
4640 LPRINT "TDRAIN(2) = ";TDRAIN(2);TAB(25);"TDRAIN(1) = ";TDRAIN(1);TAB(50);"TDRA
IN(3) = ";TDRAIN(3)
4650 LPRINT "HDRAIN(2) = ";HDRAIN(2);TAB(25);"HDRAIN(1) = ";HDRAIN(1);TAB(50);"HDRA
IN(3) = ";HDRAIN(3)
4660 LPRINT "FLODRN(2) = ";FLOEXT(2);TAB(25);"FLODRN(1) = ";FLOEXT(2)+FLOWPT+FLOSEP
(1);TAB(50);"FLODRN(3) = ";FLOEXT(2)+FLOWPT+FLOSEP(1)+FLOEXT(3)+FLOSEP(3)
4670 LPRINT "QSEP(0) = ";QSEP(0);TAB(25);"QSEP(1) = ";QSEP(1);TAB(50);"QSEP(3) = "
;QS
EP(3)
4680 LPRINT "FLOSEP(0) = ";FLOSEP(0);TAB(25);"FLOSEP(1) = ";FLOSEP(1);TAB(50);"FLOS
EP(3) = ";FLOSEP(3)
4690 LPRINT "TFRDM(2) = ";TFRDM(2);TAB(25);"TFRDM(1) = ";TFRDM(1);TAB(50);"TFRDM(3)
= ";TFRDM(3)
4700 LPRINT "HFRDM(2) = ";HFRDM(2);TAB(25);"HFRDM(1) = ";HFRDM(1);TAB(50);"HFRDM(3)
= ";HFRDM(3)
4710 LPRINT "THW = ";THW;TAB(25);"TPUMP = ";TPUMP
4720 LPRINT "HHW = ";HHW;TAB(25);"HPUMP = ";HPUMP
4730 IF DETAIL%#0 GOTO 8000
4740 LPRINT CHR$(12):LPRINT "TURBINE CYCLE SIZED FOR SPECIFIED ELECTRICAL OUTPU
T";TAB(74);CLOCK$;" ",DAY$
4750 LPRINT:LPRINT "ELECTRICAL OUTPUT = ";KWOUT;" KW(E)"
4760 LPRINT "THERMAL INPUT = ";MQADD;" KW(T)"
4770 LPRINT "CONDENSER REJECT = ";MQREJ;" KW(T)"
4780 LPRINT "GENERATOR LOSSES = ";KWOUT*(1/GEFF-1);" KW(T)":LPRINT
4790 LPRINT "STAGE";TAB(12);"STAGE FLOW";TAB(26);"LIQ. SEP";TAB(39);"VAP. SEP";
TAB(51);"EXTRACT.":LPRINT
4800 FOR I%=1 TO NS%:LPRINT I%;TAB(12);
4810 LPRINT USING "#####.###";MFLO(I%);MLSEP(I%);MVSEP(I%);

```

```

4820 IF I% = NSEXT2% THEN EXTR(I%)=MFLOEX2 ELSE 4840
4830 GOTO 4850
4840 IF I% = NSEXT3% THEN EXTR(I%)=MFLOEX3
4850 LPRINT USING "#####.###";EXTR(I%);:LPRINT " "
4860 NEXT I%
4870 LPRINT: LPRINT "MAIN VAPOR FLOW = ";MMAIN;" LB/SEC":LPRINT
4880 AA$="####.##"
4890 LPRINT "HEATER NO.";TAB(50);"1";TAB(62);"2";TAB(74);"3":LPRINT
4900 LPRINT TAB(3);"FEED FLOW, LB/SEC";TAB(45);:LPRINT USING AA$;MMAIN,MMAIN,MM
AIN
4910 LPRINT TAB(3);"FEED TEMP IN, R";TAB(45);:LPRINT USING AA$;TTO(1),TTO(2),TT
0(3)
4920 LPRINT TAB(3);"FEED ENTHALPY IN, BTU/LB";TAB(45);:LPRINT USING AA$;HTO(1),
HTO(2),HTO(3)
4930 LPRINT TAB(3);"FEED TEMP OUT, R";TAB(45);:LPRINT USING AA$;TFROM(1),TFROM(
2),TFROM(3)
4940 LPRINT TAB(3);"FEED ENTHALPY OUT, BTU/LB";TAB(45);:LPRINT USING AA$;HFROM(
1),HFROM(2),HFROM(3)
4950 LPRINT
4960 LPRINT TAB(3);"VAPOR MIX. FLOW, LB/SEC";TAB(45);:LPRINT USING AA$;MFLODR(1
),MFLODR(2),MFLODR(3)
4970 LPRINT TAB(3);"VAPOR MIX. VOL. FLOW, FT^3/SEC";TAB(45);:LPRINT USING AA$;V
INT01,VINT02,VINT03
4980 LPRINT TAB(3);"VAPOR MIX. TEMP, R";TAB(45);:LPRINT USING AA$;TOUT,TEXT(2),
TEXT(3)
4990 LPRINT TAB(3);"VAPOR MIX. QUALITY";TAB(45);:LPRINT USING AA$;XINT01,XINT02
,XINT03
5000 LPRINT TAB(3);"VAPOR MIX ENTHALPY, BTU/LB";TAB(45);:LPRINT USING AA$;HINTO
1,HINTO2,HINTO3
5010 LPRINT TAB(3);"DRAINS TEMP, R";TAB(45);:LPRINT USING AA$;TDRAIN(1),TDRAIN(
2),TDRAIN(3)
5020 LPRINT TAB(3);"DRAINS ENTHALPY, BTU/LB";TAB(45);:LPRINT USING AA$;HDRAIN(1
),HDRAIN(2),HDRAIN(3)
5030 LPRINT
5040 LPRINT TAB(3);"HEAT EXCHANGE (COND.), BTU/SEC";TAB(45);:LPRINT USING AA$;Q
HXC(1),QHXC(2),QHXC(3)
5050 LPRINT TAB(3);"LOG MEAN TEMP DIF. (COND.)";TAB(45);:LPRINT USING AA$;LMTDC
(1),LMTDC(2),LMTDC(3)
5060 LPRINT TAB(3);"HEAT EXCHANGE (D. COOL.), BTU/SEC";TAB(45);:LPRINT USING AA
$;QHxD(1),QHxD(2),QHxD(3)
5070 LPRINT TAB(3);"LOG MEAN TEMP DIF. (D. COOL.)";TAB(45);:LPRINT USING AA$;LM
TDD(1),LMTDD(2),LMTDD(3)
5080 LPRINT:LPRINT
5090 LPRINT "FEED PUMP TURBINE"
5100 LPRINT TAB(3);"VAPOR FLOW, LB/SEC";TAB(40);:LPRINT USING AA$;MFLOPT
5110 LPRINT TAB(3);"ENTHALPY IN, BTU/LB";TAB(40);:LPRINT USING AA$;H(0)
5120 LPRINT TAB(3);"ENTHALPY OUT, BTU/LB";TAB(40);:LPRINT USING AA$;HOUT
5130 LPRINT TAB(3);"POWER, KW";TAB(40);:LPRINT USING AA$;WORKT*FACTOR*3600/3412
5140 LPRINT
5150 LPRINT "FEED PUMP"

```

```

5160 LPRINT TAB(3);"FEED PUMP FLOW, LB/SEC";TAB(40);:LPRINT USING AA$;MFLOFP
5170 LPRINT TAB(3);"FEED PUMP HEAD, PSI";TAB(40);:LPRINT USING AA$;P(0)-P(NS%)+DELPYS
5180 LPRINT
5190 LPRINT "CONDENSER"
5200 LPRINT TAB(3);"VAPOR MIX. TEMP, R";TAB(40);:LPRINT USING AA$;TCDN
5210 LPRINT TAB(3);"VAPOR MIX. FLOW, LB/SEC";TAB(40);:LPRINT USING AA$;MMAIN
5220 LPRINT TAB(3);"VAPOR MIX. QUALITY";TAB(40);:LPRINT USING AA$;XTOCON
5230 LPRINT TAB(3);"VAPOR MIX. ENTHALPY, BTU/LB";TAB(40);:LPRINT USING AA$;HTOC
ON
5240 LPRINT TAB(3);"CONDENSATE TEMP, R";TAB(40);:LPRINT USING AA$;THW
5250 LPRINT TAB(3);"CONDENSATE ENTHALPY, BTU/LB";TAB(40);:LPRINT USING AA$;HHW
5260 LPRINT CHR$(12):LPRINT TAB(74);CLOCK$;"    ";DAY$
5270 LPRINT "SCHEDULE OF PIPING RUNS":LPRINT
5280 LPRINT TAB(26);"TEMP";TAB(35);"PRESS";TAB(45);"FLOW";TAB(54);"FLOW";TAB(63)
);"NO.";TAB(72);"UNIT";TAB(B1);"I.D.";TAB(90);"WALL";TAB(98);"TOTAL"
5290 LPRINT "NO.";TAB(6);"DESCRIPTION";TAB(26);"    R";TAB(35);"PSIA";TAB(45);"LB
/S";TAB(54);"CF/S";TAB(63);"LINES";TAB(72);"L,FT";TAB(B1);"INCH";TAB(90);"INCH"
;TAB(98);"LBS"
5300 LPRINT
5310 FOR I%=1 TO 23
5320 LPRINT I%;TAB(6);LBL$;(I%);TAB(24);
5330 LPRINT USING "####.### ";TL(I%);PL(I%);MF(I%);CF5(I%);NUM%(I%);LG(I%);ID(I%
);WALL(I%);WT(I%)
5340 NEXT I%
5350 LPRINT:LPRINT TAB(77);"TOTAL PIPING WEIGHT";:LPRINT USING "####.###";TOTWT
5360 LPRINT:LPRINT:LPRINT "DESIGN CHARACTERISTICS OF FEED HEATERS":LPRINT "(USI
ND";ODTUBE*12;"IN. OD X";WALLT;"IN. WALL TUBING)"
5370 LPRINT:LPRINT TAB(6);"NO.";TAB(14);"PITCH,";TAB(23);"SHELL";TAB(32);"WALL"
;TAB(41);"LENGTH";TAB(76);"WT/";TAB(85);"TOTAL"
5380 LPRINT "NO.";TAB(6);"TUBES";TAB(14);"IN."/TAB(23);"DIA,IN";TAB(32);"IN";TA
B(41);"FT";TAB(51);"HO";TAB(60);"UIC";TAB(68);"UID";TAB(76);"HTR,LB";TAB(85);"W
T,LB";TAB(94);"SIGMA"
5390 LPRINT
5400 AA$="####.###":BB$="#####. "
5410 FOR I%=1 TO NH%
5420 LPRINT I%;TAB(6);NTUBES(I%);TAB(11);
5430 LPRINT USING AA$;PITCH(I%)*12;DSHELL(I%)*12;WALLSH(I%)*12;LENGTH(I%);:LPRI
NT USING BB$;HO(I%);UIC(I%);UID(I%);
5440 LPRINT USING AA$;WTHTR(I%);WTHTR(I%)*NUM%(2);:LPRINT USING BB$;SIGMAH(I%)
5450 NEXT I%
5460 LPRINT:LPRINT TAB(62);"TOTAL HEATER WEIGHT ";:LPRINT USING AA$;TOTHTR
5999 GOTO 8000
6000 REM *****SUBROUTINE KTHERMO*****
6010 REM SUBROUTINE RETURNS THERMODYNAMIC PROPERTIES OF POTASSIUM FROM T
6020 P=EXP(14.10927-18717.2/T-.53299*LOG(T)) '[ATMOSPHERES]
6030 VF=1/(52.768-.0074975*(T-460)-5.255E-07*(T-460)^2+4.98E-11*(T-460)^3) '[FT
^3/LB]
6040 B=-1*ABS(EXP(-8.931+11261.2/T+LOG(T)))
6050 B1=-1*ABS(EXP(-8.931+11261.2/(2+T)+LOG(2+T)))

```

```

6060 DBDT=(B1-B)/2!
6070 C=EXP(1.35231+14703.6/T)
6080 C1=EXP(1.35231+14703.6/(T+2))
6090 DCDTX=(C1-C)/2!
6100 D=-1*ABS(EXP(3.3606+18107.1/T))
6110 D1=-1*ABS(EXP(3.3606+18107.1/(2+T)))
6120 DDDT=(D1-D)/2!
6130 E=0!
6140 V1=.73*T/P
6150 N1=0
6160 MUD=P*V1/(.73*T)-1-B/V1-C/V1^2-D/V1^3
6170 N1=N1+1
6180 IF N1>10 THEN 6330
6190 SLOPE=P/(.73*T)+B/V1^2+2*C/V1^3+3*D/V1^4
6200 V2=V1-MUD/SLOPE
6210 IF ABS(V1-V2)<.01 THEN 6240
6220 V1=V2
6230 GOTO 6160
6240 VG=V2 'IFT^3/LB-MOLE]
6250 HFG=2.72*P*(18717!/T-.53299)*(VG/39.1-VF)
6260 HG0=998.95+.127*T+24836!*EXP(-39375!/T)
6270 HG=HG0+1.987*(T/39.1)*((B-T*DBDT)/VG+(C-T*DCDTX/2)/VG^2+(D-T*DDDT/3)/VG^3)
6280 HF=HG-HFG
6290 SFG=HFG/T
6300 SG0=.127*LOG(T)+.18075+.7617*EXP(-31126/T)
6310 SG=SG0-(1.987/39.1)*(LOG(P)-LOG(P*VG/(.73*T))+B/VG+T*DBDT/VG+C/(2*VG^2)+T*
DCDTX/(2*VG^2)+D/(3*VG^3)+T*DDDT/(3*VG^3))
6320 SF=SG-SFG:VG=VG/39.1:GOTO 6350
6330 LPRINT "*****SOLUTION FOR VG WON'T CONVERGE FOR TEMP = ";T;"*****"
6340 T=T+1:GOTO 6020
6350 RETURN
6450 REM *****VFFROMT*****
6460 VF=1/(52.768-.0074975*(T-460)-5.255E-07*(T-460)^2+4.98E-11*(T-460)^3) 'IFT
^3/LB]
6470 RETURN
6500 REM *****KTFROMP*****
6510 REM CALCULATES SATURATION TEMPERATURE (R) FROM GIVEN PRESSURE (ATM)
6520 T1=1000
6530 E=18717.2/T1+.53299*LOG(T1)-14.10927+LOG(P)
6540 M=-18717.2/T1^2+.53299/T1
6550 T2=T1-E/M
6560 IF ABS(T1-T2)<.01 THEN 6590
6570 T1=T2
6580 GOTO 6530
6590 TEMP=T2
6600 RETURN
7000 *****SUBROUTINE TFROMH*****
7010 'CALCULATES SATURATION TEMP (R) FROM HF
7020 GOSUB 6000
7030 HA=HF:TA=T:RES=HA-H

```

```

7040 IF ABS(RES)<.1 GOTO 7110
7050 T=T+10
7060 GOSUB 6000
7070 HB=HF:TB=T
7080 M=(HB-HA)/(TB-TA)
7090 T=TA-RES/M
7100 GOTO 7020
7110 TEMP=TA
7120 RETURN
7200 '***** POTASSIUM TRANSPORT PROPERTIES SUBROUTINE *****
7210 TF=TR-460!
7220 TC=(TR-492)/1.8
7230 IF TR<1158 THEN MU=EXP(1189.98/TR-1.6286) ELSE MU=EXP(1698.156/TR-2.0675)
7240 K=57.82*(.438-.000222*TC+39.5/(TC+273.2))
7250 CP=.239*(.84074-3.1688E-04*TC+3.1435E-07*TC^2)
7260 RHOFL=52.768-.0074975*TF-5.255E-07*TF^2+4.98E-11*TF^3
7270 RETURN
7500 DATA 2500.,1.00,1800.,9,.95,.85,5.
7510 'TBOIL,XBOIL,TCON,NS%,GEFF,DEFF,EXLOSS
7520 DATA 0,0,0,0,2,0,0,1,0
7530 'NS% VALUES FOR SEP% (0=NO SEP;1=INTERSTAGE SEP;2=INTERSPOOL SEP)
7540 DATA 0.7,0.6,3,350.,4.0,115.,10.,10.,20,100.
7550 'PTEFF,PEFF,NHX,DELPTPT,JETRAT,DELPSYS,SCCON,TTD,DCDT,BFSC
7560 DATA 1000,1
7570 'KWOOUT,DETAIL%(0=NO,1=YES)
7580 'THE FOLLOWING DATA IS NOT NEEDED IF [DETAIL] IS [0]
7590 'FOLLOWING ARE 23 SETS OF LLBL$ (LABEL),LG (LENGTH,FT),TY (TYPE FLOW;1=VAP
OR,2=MIX,3=LIQUID), NUM (NUMBER OF LINES)
7600 DATA "REACTOR OUTLET",6,1,1,"TURBINE INLET",3,1,2,"TURBINE OUTLET",3,1,2
7610 DATA "CONDENSER INLET",3,1,2,"CONDENSER OUTLET",3,3,2,"COND.JET OUTLET",4,
3,2
7620 DATA "HTR 3 FEED",3,3,2,"HTR 1 FEED",2,3,2,"HTR 2 FEED",2,3,2
7630 DATA "BOILER FEED",6,3,1,"P.TURB. INLET",3,1,2,"P.TURB. OUTLET",4,1,2
7640 DATA "HTR 2 EXTR.",4,1,2,"HTR 2 DRAIN",4,3,2,"HTR 1 SEPS",4,2,2
7650 DATA "HTR 1 MIX",2,1,2,"HTR 1 DRAIN",4,3,2,"HTR 3 EXTR.",4,1,2
7660 DATA "HTR 3 SEPS",4,2,2,"HTR 3 MIX",2,2,2,"HTR 3 DRAIN",4,3,2
7670 DATA "COND. SEPS",3,2,2,"COND.JET RECIRC.",4,3,2
7680 'FOLLOWING ARE NPTA,NPTB,TMAT,VELV,VELM,VELL,RHOA,RHOB,KA,KB
7690 'ALLOY A IS HI-TEMP MATL, B IS LO-TEMP, TMAT IS MATL SWITCH TEMP
7700 DATA 9,5,2070,450,100,10,.604,.31,31.0,31.0
7710 'FOLLOWING ARE NPTA PAIRS OF TEMP AND STRESS FOR HI-TEMP ALLOY
7720 DATA 1859,56560,1960,46410,2059,36260,2160,31330,2259,26830,2360,22630,245
8,19870,2560,15520,2659,10150
7730 'FOLLOWING ARE NPTB STRESS VALUES FOR LO-TEMP ALLOY THAT GO WITH FIRST NPT
B TEMPS FROM STATEMENT 7720
7740 DATA 7250,5950,5080,4350,3050
7750 'FOLLOWING ARE HEATER TUBE OD (IN.) AND WALL THICKNESS (IN.)
7760 DATA .25,.020
8000 ZZ=TCON 'A DUMMY STATEMENT TO PREVENT CLEARING OUT THE REGISTERS AT END

```

SUBROUTINES FOR PROPERTIES OF SODIUM

```

6000 *****SUBROUTINE NATHERMO*****
6010 'CALCULATION OF SATURATION PROPERTIES (BASED ON GOLDEN & TOKAR, ANL-7323)
6020 IF T<=2059.7 GOTO 6030 ELSE GOTO 6050
6030 P=(3032660!/T^.5)*EXP(-23073.3/T)      'ATMOSPHERES
6040 GOTO 6060
6050 P=(6881760.2#/(T^.61344))*EXP(-22981.96/T) 'ATMOSPHERES
6060 HF=.389352*T-5.529955E-05*T^2+1.13726E-08*T^3-29.02    'BTU/LB
6070 SF=.389344*LOG(T)-1.10557E-04*T+1.70408E-08*T^2-1.792026
6080 VF=1/(59.566-.0079504*(T-459.7)-2.872E-07*(T-459.7)^2+6.035E-11*(T-459.7)^3)
6090 TWO=EXP(-9.958451+16588.3/T)
6100 FOUR=EXP(-24.59115+37589.7/T)
6110 U=P*TWO
6120 W=(P^3)*FOUR
6130 X1=.8
6140 X1P=X1
6150 X1=X1-(W*X1^4+U*X1^2+X1-1!)/(4!*W*X1^3+2!*U*X1+1!)
6160 IF ABS(X1-X1P)/X1<=.000001 GOTO 6170 ELSE GOTO 6140
6170 X2=U*X1^2
6180 X4=1!-X1-X2
6190 ABAR=22.991*(X1+2!*X2+4!*X4)
6200 VG=.730229*T/(ABAR*P)
6210 Z=22.991/ABAR
6220 DHF1=25980.7-2.21312*T+7.06278E-04*T^2-1.4526E-07*T^3
6230 DHF2=2!*DHF1-18304!
6240 DHF4=4!*DHF1-41478!
6250 HFG=1.8*(X1*DHF1+X2*DHF2+X4*DHF4)/ABAR
6260 HG=HF+HFG
6270 SG=SF+HFG/T
6275 SFG=SG-SF
6280 RETURN
6450 REM *****VFROMT*****
6460 VF=1/(59.566-.0079504*(T-459.7)-2.872E-07*(T-459.7)^2+6.035E-11*(T-459.7)^3)
6470 RETURN
6500 REM *****NATFRDM*****
6510 REM CALCULATES SATURATION TEMPERATURE (R) FROM GIVEN PRESSURE (ATM)
6520 T1=2600
6530 E=6881760.2#*EXP(-22981.96/T1)/T1^.61344-P
6540 M=6881760.2#*EXP(-22981.96/T1)*(22981.96*T1^-2.61344-.61344*T1^-1.61344)
6550 T2=T1-E/M
6560 IF ABS(T1-T2)<.01 THEN 6590
6570 T1=T2

```

```
6580 GOTO 6530
6590 TEMP=T2
6600 RETURN
7000 *****SUBROUTINE TFROMH*****
7010 'CALCULATES SATURATION TEMP (R) FROM HF
7020 GOSUB 6000
7030 HA=HF:TA=T:RES=HA-H
7040 IF ABS(RES)<.1 GOTO 7110
7050 T=T+10
7060 GOSUB 6000
7070 HB=HF:TB=T
7080 M=(HB-HA)/(TB-TA)
7090 T=TA-RES/M
7100 GOTO 7020
7110 TEMP=TA
7120 RETURN
7200 ***** SODIUM TRANSPORT PROPERTIES SUBROUTINE *****
7210 TF=TR-460!
7220 TC=(TR-492)/1.8
7230 MU=EXP(2.3493+914.52/TR-.4925*LOG(TR)) '(EQ. 5.19A,P.50,GOLDEN & TOKAR)
7240 K=54.306-.01878*TF+2.0914E-06*TF^2 '(EQ. 5.9,P.43,GOLDEN & TOKAR)
7250 CP=.389352-1.10599E-04*TR+3.41178E-08*TR^2 '(EQ. 5.4,P.40, GOLDEN & TOKAR
)
7260 RHOFL=59.566-.0079504*TF-2.872E-07*TF^2+6.035E-11*TF^3
7270 RETURN
```

TURBINE EXPANSION LINE CALCULATION

15:24:21 06-21-1985

THROT. T, R = 2500 COND. T, R = 1800
 NO. OF STAGES = 9 DRY STAGE EFF, % = 85
 EXHAUST LOSS, BTU/LBM = 5

STAGE	EXIT T	EXIT P	EXIT S	EXIT H	EXIT X	FLOW RATIO	DELTA H
0	2500.0	170.68640	1.01978	1218.79800	1.00000	1.00000	0.00000
1	2422.2	136.49230	1.02134	1199.39100	0.97965	1.00000	19.48625
2	2344.4	107.48840	1.02319	1179.98300	0.95963	1.00000	19.48845
3	2266.7	83.20783	1.02538	1160.55900	0.93989	1.00000	19.42444
4	2188.9	63.21457	1.02794	1141.10200	0.92039	1.00000	19.45654
5	2102.9	45.52976	1.06554	1191.64600	0.99049	0.98217	19.58769
6	2027.2	33.29497	1.06888	1168.11800	0.96524	1.00000	23.52747
7	1951.5	23.74523	1.07101	1144.72700	0.94087	1.00000	23.39124
8	1875.7	16.46483	1.08298	1137.48600	0.93638	0.97418	23.29480
9	1800.0	11.06021	1.08984	1117.98800	0.91834	1.00000	19.57825

MOISTURE SEPARATIONS (LIQ. OUT AND VAP. OUT ARE PER LBM ENTERING STAGE)

STAGE	TYPE	LIQ. OUT	LIQ. H	VAP. OUT	VAP. H	VAP. S	ELEP X
5	INTERSPOOL	0.00893	410.97160	0.00089	1121.59400	1.03091	0.98118
8	INTERSTAGE	0.02066	364.39910	0.00516	1121.43200	1.07462	0.91737

LAST STAGE ELEP: H = 1112.900 S = 1.007061 X = .9123622

TURBINE CYCLE CALCULATION (PER LB PRIME VAPOR)

15:24:21 06-21-1985

NO. OF HEATER STAGES = 3 PUMP TURBINE DELTA T = 350
 JET PUMP FLOW RATIO = 4 PUMP TURB. EFF. = .7 PUMP EFF. = .6 GEN. EFF. = .95
 SYSTEM PRESSURE DROP = 115 CONDENSER SUBCOOLING = 10
 HEATER TERMINAL TEMP. DIFF. = 10 DRAIN COOLER DELTA T = 20
 BOILER FEED SUBCOOLING = 100

*** THREE FEED HEATERS FED FROM STAGES 1 AND 6 AND FROM PUMP TURBINE EXHAUST ***

STAGE	STAGE FLOW	STAGE WORK	LIQ. SEP	VAP. SEP	STAGE EFF.
1	0.96666	18.75916	0.00000	0.00000	0.83729
2	0.90087	17.48452	0.00000	0.00000	0.81675
3	0.90087	17.49893	0.00000	0.00000	0.79652
4	0.90087	17.52785	0.00000	0.00000	0.77654
5	0.90087	17.57393	0.00012	0.00001	0.75681
6	0.81274	19.12175	0.00000	0.00000	0.82474
7	0.78170	18.28483	0.00000	0.00000	0.79945
8	0.78170	18.20945	0.01615	0.00404	0.77502
9	0.76151	14.90904	0.00000	0.00000	0.76999

HEAT ADDED = 750.8926 HEAT REJECTED = 591.5229
 TOTAL TURBINE WORK = 159.3695 CYCLE EFFICIENCY = .21224 PLANT EFF. = .201628

PUMP TURBINE FLOW = 3.334435E-02 PUMP TURBINE EXH. ENTHALPY = 1143.153
 PUMP TURBINE WDRK = 2.522322 BOILER FEED TEMPERATURE = 2400 DEG. R
 EXTRACTION FLOW TO HEATER 2 = 6.578374E-02 EXTRACTION FLOW TO HEATER 3 = 3.104579E-02

TEXT(2)= 2422.222	TOUT= 2150	TEXT(3)= 2027.199
HEXT(2)= 1199.391	HOUT= 1143.153	HEXT(3)= 1168.118
FLOEXT(2)= 6.578374E-02	FLOWPT= 3.334435E-02	FLOEXT(3)= 3.104579E-02
TDRAIN(2)= 2160	TDRAIN(1)= 2024.466	TDRAIN(3)= 1822.769
HDRAIN(2)= 420.6129	HDRAIN(1)= 393.787	HDRAIN(3)= 353.9194
FLDRN(2)= 6.578374E-02	FLDRN(1)= 9.912800E-02	FLDRN(3)= .2183042
QSEP(0)= 10.41174	QSEP(1)= 0	QSEP(3)= 41.91243
FLOSEP(0)= .0201854	FLOSEP(1)= 0	FLOSEP(3)= 8.813020E-02
TFROM(2)= 2400	TFROM(1)= 2140	TFROM(3)= 2004.466
HFRDM(2)= 467.9051	HFRDM(1)= 416.6741	HFRDM(3)= 389.9223
THW = 1790	TPUMP = 1802.769	
HHW = 347.4491	HPUMP = 349.9714	

TURBINE CYCLE SIZED FOR SPECIFIED ELECTRICAL OUTPUT

15:24:21 06-21-1985

ELECTRICAL OUTPUT = 1000 KW(E)
 THERMAL INPUT = 4959.629 KW(T)
 CONDENSER REJECT = 3986.995 KW(T)
 GENERATOR LOSSES = 52.63162 KW(T)

STAGE	STAGE FLOW	LIQ. SEP	VAP. SEP	EXTRACT.
1	6.051	0.000	0.000	0.412
2	5.648	0.000	0.000	0.000
3	5.648	0.000	0.000	0.000
4	5.648	0.000	0.000	0.000
5	5.640	0.502	0.050	0.000
6	5.088	0.000	0.000	0.194
7	4.893	0.000	0.000	0.000
8	4.893	0.101	0.825	0.000
9	4.767	0.000	0.000	0.000

MAIN VAPOR FLOW = 6.260051 LB/SEC

HEATER NO.	1	2	3
FEED FLOW, LB/SEC	6.260	6.260	6.260
FEED TEMP IN, R	2004.466	2148.000	1882.769
FEED ENTHALPY IN, BTU/LB	389.922	416.674	349.971
FEED TEMP OUT, R	2148.000	2400.000	2004.466
FEED ENTHALPY OUT, BTU/LB	416.674	467.985	389.922
VAPOR MIX. FLOW, LB/SEC	0.621	0.412	1.367
VAPOR MIX. VOL. FLOW, FT^3/SEC	1.854	1.659	3.667
VAPOR MIX. TEMP, R	2150.000	2422.222	2027.199
VAPOR MIX. QUALITY	0.313	0.980	0.178
VAPOR MIX ENTHALPY, BTU/LB	663.658	1199.391	536.925
DRAINS TEMP, R	2024.466	2168.000	1822.769
DRAINS ENTHALPY, BTU/LB	393.787	420.613	353.919
HEAT EXCHANGE (COND.), BTU/SEC	152.043	299.420	194.775
LOG MEAN TEMP DIF. (COND.)	47.544	97.939	75.954
HEAT EXCHANGE (O. COOL.), BTU/SEC	15.425	21.288	55.328
LOG MEAN TEMP DIF. (O. COOL.)	59.657	94.885	72.769

FEED PUMP TURBINE

VAPOR FLOW, LB/SEC	8.209
ENTHALPY IN, BTU/LB	1218.798
ENTHALPY OUT, BTU/LB	1143.153
POWER, KW	16.660

FEED PUMP

FEED PUMP FLOW, LB/SEC	7.825
FEED PUMP HEAD, PSI	274.626

CONDENSER

VAPOR MIX. TEMP, R	1800.000
VAPOR MIX. FLOW, LB/SEC	6.260
VAPOR MIX. QUALITY	0.705
VAPOR MIX. ENTHALPY, BTU/LB	938.972
CONDENSATE TEMP, R	1798.000
CONDENSATE ENTHALPY, BTU/LB	347.449

15:24:21 06-21-1985

SCHEDULE OF PIPING RUNS

NO.	DESCRIPTION	TEMP R	PRESS PSIA	FLOW LB/S	FLOW CF/S	NO. LINES	UNIT L,FT	I.D. INCH	WALL INCH	TOTAL LBS
1	REACTOR OUTLET	2500.000	170.686	6.260	21.002	1.000	6.000	2.925	0.020	8.047
2	TURBINE INLET	2500.000	170.686	6.051	20.301	2.000	3.000	2.034	0.020	5.611
3	TURBINE OUTLET	1800.000	11.060	4.767	181.517	2.000	3.000	6.001	0.020	8.556
4	CONDENSER INLET	1800.000	11.060	6.260	182.897	2.000	3.000	6.104	0.020	8.588
5	CONDENSER OUTLET	1790.000	11.060	6.260	0.149	2.000	3.000	1.169	0.020	1.668
6	COND.JET OUTLET	1790.000	21.060	7.825	0.186	2.000	4.000	1.307	0.020	2.482
7	HTR 3 FEED	1802.769	285.686	6.260	0.149	2.000	3.000	1.169	0.021	1.724
8	HTR 1 FEED	2004.466	285.686	6.260	0.156	2.000	2.000	1.196	0.031	1.777
9	HTR 2 FEED	2140.000	285.686	6.260	0.161	2.000	2.000	1.214	0.020	2.248
10	BOILER FEED	2400.000	285.686	6.260	0.171	1.000	6.000	1.771	0.020	4.893
11	P.TURB. INLET	2500.000	170.686	0.209	0.700	2.000	3.000	0.378	0.020	1.087
12	P.TURB. OUTLET	2150.000	54.676	0.209	1.834	2.000	4.000	0.611	0.020	2.300
13	HTR 2 EXTR.	2422.222	136.492	0.412	1.659	2.000	4.000	0.581	0.020	2.191
14	HTR 2 DRAIN	2160.000	136.492	0.412	0.011	2.000	4.000	0.312	0.020	1.210
15	HTR 1 SEPS	0.000	0.000	0.000	0.000	2.000	4.000	0.000	0.020	0.000
16	HTR 1 MIX	2150.000	54.676	0.621	1.854	2.000	2.000	0.615	0.020	1.156
17	HTR 1 DRAIN	2024.466	54.676	0.621	0.016	2.000	4.000	0.377	0.020	0.743
18	HTR 3 EXTR.	2027.199	33.295	0.194	2.810	2.000	4.000	0.757	0.020	1.452
19	HTR 3 SEPS	2102.932	45.530	0.552	0.570	2.000	4.000	0.723	0.020	2.700
20	HTR 3 MIX	2027.199	33.295	1.367	3.667	2.000	2.000	1.033	0.020	1.733
21	HTR 3 DRAIN	1822.769	33.295	1.367	0.034	2.000	4.000	0.560	0.020	1.085
22	COND. SEPS	1875.733	16.465	0.126	0.682	2.000	3.000	0.790	0.020	1.137
23	COND.JET RECIRC.	1802.769	285.686	1.565	0.037	2.000	4.000	0.585	0.020	1.130

TOTAL PIPING WEIGHT 63.525

DESIGN CHARACTERISTICS OF FEED HEATERS
(USING .25 IN. OD X .02 IN. WALL TUBING)

NO.	NO. TUBES	PITCH, IN.	SHELL DIA,IN	WALL IN	LENGTH FT	HO	UIC	UID	WT/ HTR,LB	TOTAL WT,LB	SIGMA
1	9	0.448	1.531	0.020	5.477	2294.	2407.	1458.	9.936	19.872	31756.
2	8	0.473	1.521	0.020	7.114	1832.	1970.	1206.	12.050	24.100	20826.
3	17	0.360	1.701	0.020	1.825	4648.	4349.	2122.	2.643	5.286	5321.

TOTAL HEATER WEIGHT 49.259

DISTRIBUTION

Internal Distribution

- | | |
|---|---|
| <ol style="list-style-type: none"> 1. D. E. Bartine 2. R. H. Cooper 3. W. W. Engle, Jr. 4. R. L. Graves 5. H. W. Hoffman 6. R. S. Holcomb 7. J. E. Jones Jr. 8. D. B. Lloyd 9. R. E. MacPherson <p style="margin-top: 10px;">10-14. J. C. Moyers</p> <p style="margin-top: 10px;">15. F. R. Mynatt</p> | <ol style="list-style-type: none"> 16. J. P. Nichols 17. H. Postma 18. W. L. Roberts 19. H. E. Trammell 20. C. D. West 21. G. L. Yoder 22. ORNL Patent Office 23. Central Research Library 24. Document Reference Section <p style="margin-top: 10px;">25-26. Laboratory Records Department</p> <p style="margin-top: 10px;">27. Laboratory Records (RC)</p> |
|---|---|

External Distribution

28. S. K. Bhattacharyya, Argonne National Laboratory, 9700 S. Cass Avenue, Argonne, IL 60439
29. B. W. Colston, MS F671, Los Alamos National Laboratory, P. O. Box 1663, Los Alamos, NM 87545
30. C. M. Cox, Hanford Engineering Development Laboratory, P. O. Box 1970, Richland, WA 99352
31. L. Cropp, Sandia National Laboratories, P. O. Box 5800, Albuquerque, NM 87185
32. P. Dirkmaat, DOE-ID, 550 Second Street, Idaho Falls, ID 83274
33. E. U. Khan, U.S. Dept. of Energy, NE-54, Germantown, Washington, D.C. 20545
34. E. T. Mahefkey, Jr., Aeropropulsion Laboratory (POOC-5), U.S. Air Force Wright Aeronautical Laboratories, Wright-Patterson Air Force Base, OH 45433

35. J. R. Powell, Brookhaven National Laboratory, Associated Universities, Inc., Upton, NY 11973
36. J. M. Smith, NASA Lewis Research Center, MS 500-202, Building 4013, 21000 Brookpark Road, Cleveland, OH 44135
37. R. J. Sovie, NASA Lewis Research Center, MS 500-202, Building 4013, 21000 Brookpark Road, Cleveland, OH 44135
38. R. Verga, Strategic Defense Initiative/Director of Energy, The Pentagon, Washington, D.C. 20301-7100
39. E. J. Wahlquist, U.S. Dept. of Energy, NE-54, Germantown, Washington, D.C. 20545
40. C. E. Walter, Lawrence Livermore National Laboratory, University of California, P. O. Box 808, Livermore, CA 94550
41. R. D. Widrig, Pacific Northwest Laboratory, Battelle Boulevard, Richland, WA 99532
42. R. Wylie, Strategic Defense Initiative/Director of Energy, The Pentagon, Washington, D.C. 20301-7100
43. H. R. Zweig, Energy Technology Engineering Center, Rocketdyne Division, Rockwell International Corp., P. O. Box 1449, Canoga Park, CA 91304
44. Office of Assistant Manager for Energy, Research and Development, DOE-Oak Ridge Operations, Oak Ridge, TN 37831
- 45-71. Office of Scientific and Technical Information, P. O. Box 62, Oak Ridge, TN 37831